
Development of Parameterized Verilog‑AMS Model 
of Photovoltaic Cells 

Elissaveta Gadjeva1 and Georgi Valkov2 

Abstract – A parameterized Verilog-AMS model of 
photovoltaic cell, module and panel is developed. The model is 
available in two variants: one that includes a detailed Spice diode 
description and the other which is simplified and includes only 
the diode equations that affect the behavior of the PV elements. 
Definitions for obtaining the model characteristics from the 
simulation results are provided. The model is realized in the 
Dolphin Integration SMASH environment.  
 

Keywords – Photovoltaic cells, Parameterized models, Verilog-
AMS language. 

I. INTRODUCTION 

The design and optimization of photovoltaic systems 
require development of computer models of PV cells and 
panels. These models are of significant importance for the 
simulation of maximum power point tracking (MPPT) control 
systems and allow the investigation and optimization of the 
dynamics of PV power systems. 

Photovoltaic models of PV cells and modules are developed 
in [1-3]. Methods for parameter extraction and simulation of 
photovoltaic modules by taking the manufacturer specified 
data are developed in [4-6]. A sun-tracking system based on 
the Verilog HDL and using an FPGA chip as controlling 
platform is proposed in [7]. Combining all analog and digital 
blocks of a project, using a behavior HDL like Verilog-AMS 
and VHDL-AMS [9] allow the entire project to be simulated 
and verified as a whole. A SoC design of solar irrigation 
control system based on FPGA is offered in [8]. 

Parameterized Verilog-AMS macromodel is created in the 
present paper that can represent a single PV cell, as well as an 
entire panel consisting of PV cells connected in series and 
parallel. An additional model using a simplified version of the 
diode is also described. Definitions for obtaining the model 
characteristics from the simulation results are provided, 
according to the pattern file syntax used by SMASH [10].  

II. MODELS OF PV CELL AND PANEL 

A. Model of a single PV cell 

The schematic representation of the macromodel for a 
single PV cell is provided in Fig. 1. I1 is a DC current source 
with value I1=ICC.KSH, where ICC is a function of the solar 

irradiation and the shading factor KSH models the behavior of 
the cell under reduced light conditions. Its range is [0, 1], 
where 1 is for unshaded cell. R1 and R2 model the shunt and 
series resistance losses. Their values are R1=RSH and R2=RS. 
The diode models the non-linear behavior, its parameters are 
IS and N, where IS is the saturation current and N is the 
emission coefficient. 

 
Fig. 1. Parameterized model of a single solar cell. 

A generic symbol for the PV cell model is shown in Fig. 2. 
 

 
Fig. 2. Generic symbol of PV cell. 

B. Model of PV panel 

The model of PV panel uses the same schematic 
representation as the single PV cell shown in Fig. 1, but with 
scaled parameters to represent a whole panel. A panel consists 
of NP modules connected in parallel, where each module has 
NS cells connected in series, as shown in Fig. 3. 

 

 
Fig. 3. Structure of PV panel. 

In the case when NS=1 and NP=1, the model represents a 
single PV cell. The values for model parameters are 
I1=ICC.KSH.NP, R1=RSH.NP, R2=RS.NS, IS=IS.NP, N=N.NS. 

A generic symbol for the PV panel model is shown in 
Fig. 4. 

 
Fig. 4. Generic symbol of PV panel. 

III. VERILOG-AMS IMPLEMENTATION  
OF PV PANEL 

A. Verilog-AMS code of PV panel with full diode description 

The Verilog-AMS implementation of the model has the 
following form: 

 

module pv_cell_01_vams(n1, n2); 
    inout       n1, n2; 
    electrical  n1, n2, n3; 
    parameter real Icc=  4.0; 
    parameter real Ksh=  1.0; 
    parameter real Rs =  1.0m; 
    parameter real Rsh=100.0; 
    parameter real Ns =  1.0; 
    parameter real Np =  1.0; 
    parameter real Is =  1.0u; 
    parameter real N  =  1.5; 
analog begin 
    I(n3, n1)<+ V(n3, n1)*Np/Rsh - Icc*Ksh*Np; 
    I(n3, n2)<+ V(n3, n2)*Np/(Ns*Rs); 
end 
    diode_pv_sp #(.Is(Is*Np), .N(N*Ns)) 
            D_pv (.n2(n3), .n1(n1)); 
endmodule  

 

The block has two electrical terminals, n2 is the positive one 
and n1 is the negative. The model parameters are defined. The 
elements I1, R1 and R2 are modeled by their component 
equations. A diode block is added between nodes n3 and n1, to 
model the non-linear behavior. 

SMASH is a mixed language simulator and allows 
subcircuits defined using the Spice language syntax to be 
added to Verilog-AMS nets and vice versa. The diode_pv_sp 
block wraps a Spice diode model inside a subcircuit, using the 
following Spice description: 
 

.SUBCKT diode_pv_sp n1 n2  PARAMS: Is=1 N=1 
    .MODEL  diode_pv D (Is=Is N=N) 
    D1      n2  n1   diode_pv 
.ENDS  

 

B. Model of PV panel with simplified diode description 

For the purpose of simulating PV cells behavior, the diode 
is only used in forward direction. Additional effects like 
junction capacitance can also be omitted. Hence it is possible 
to simplify the model of PV panel with Spice diode, by 
replacing the diode with its equivalent equation. The model is 
described by the following Verilog-AMS code: 

 
 

module pv_cell_02_vams(n1, n2); 
    inout       n1, n2; 
    electrical  n1, n2, n3; 
    parameter real Icc  =   4.0  from (0:inf); 
    parameter real Ksh  =   1.0  from [0:1]; 
    parameter real Rs   =   1.0m from (0:inf); 
    parameter real Rsh  = 100.0  from (0:inf); 
    parameter real Ns   =   1.0  from [1:inf); 
    parameter real Np   =   1.0  from [1:inf); 
    parameter real Is   =   1.0u from (0:inf); 
    parameter real N    =   1.5  from (0:inf); 
    parameter real Tn   = 300.15 from (0:inf); 
    parameter real XTI  =   3.0  from (0:inf); 
    parameter real Eg   =   1.11 from (0:inf); 
analog begin 
    I(n3, n1) <+ V(n3, n1)*Np/Rsh - Icc*Ksh*Np 
    + Np*Is*(exp(V(n3, n1)/(Ns*N*$vt))-1) 
    * exp(($temperature/Tn-1)*Eg/(Ns*N*$vt)) 
    * pow($temperature/Tn, XTI/(Ns*N)); 

    I(n3, n2) <+ V(n3, n2)*Np/(Ns*Rs); 
end 
endmodule  

IV. SIMULATED CHARACTERISTICS IN DOLPHIN 
INTEGRATION SMASH 

The characteristics for the Verilog-AMS model of PV 
panel with full diode description are obtained for a single cell 
using DC sweep as the primary simulation analysis. To 
determine the dependence on RS, RSH, KSH and temperature, the 
following parametric sweeps are run together with the primary 
analysis: RS: value list: 0.1m, 1m, 10m, 100m; RSH: value list: 1, 
10, 100; KSH: linear from 0.25 to 1, with a step size of 0.25 and 
TEMPER: linear from 0 to 75, with a step size of 25. The 
remaining model parameters are given in Fig. 4. 

The current and power characteristics are obtained using 
the waveforms I(V1) and IN(V1.POWER). The following 
characteristics are also obtained from the simulation results: 
current and voltage at the maximum power point IPP and VPP, 
fill factor FF, maximum power PMAX, short circuit current ISC, 
and open circuit voltage VOC. The parameter FF is defined as 
follows: 

 MAX

T

P
FF

P
 ,  (1) 

where:  maxMAXP P ,  (2) 

 T SC OCP I V .  (3) 
The definitions for obtaining PMAX, ISC, VOC, PT, FF, VPP, 

and IPP from the simulation results are created using the 
.MEASURE command according to the pattern file syntax 
used by SMASH: 

 
 

.MEASURE ANALYSIS=DC NAME=Pmax 
+   WAVEFORM=IN(V1.POWER) EXTRACT=MAX 
.MEASURE ANALYSIS=DC NAME=Isc 
+   WAVEFORM=I(V1) EXTRACT=MAX 
.MEASURE ANALYSIS=DC NAME=Voc 
+   WAVEFORM=I(V1) EXTRACT=X ATY=0 
+   CROSS=DOWN OCCUR=1 
.MEASURE ANALYSIS=DC NAME=PT VALUE={Isc*Voc} 
.MEASURE ANALYSIS=DC NAME=FF VALUE={Pmax/PT} 
.MEASURE ANALYSIS=DC NAME=Vpp 
+   WAVEFORM=IN(V1.POWER) EXTRACT=ATMAX ATY=0 
.MEASURE ANALYSIS=DC NAME=Ipp 
+   WAVEFORM=I(V1) EXTRACT=Y ATX=Vpp  

 

The current and power characteristics as a function of RS 
are shown in Fig. 5.  

Fig. 5. Current and power characteristics as a function of RS: I(V1,RS) 
and P(V1,RS). 

High values for the series resistance tend to cause a 
significant drop in the output power of the PV elements as 
well as reduction of IPP, VPP and FF (Fig. 6). 

The current and power characteristics as a function of RSH 
are shown in Fig. 7. High values for the shunt resistance have 
no impact on the performance, while low values act like an 
additional consumer, reducing the power at the output of the 
PV elements and also reducing IPP, VPP and FF (Fig. 8). The 
shading factor KSH indicates what part of the PV cell or panel 
is not shaded. The output current is proportional to the solar 
irradiation and hence proportional to KSH. Lower values for 
KSH result in a lower output power (Fig. 9), and also cause a 
reduction in IPP, VPP and FF, as shown in Fig. 10. 

The temperature dependences of the output current and 
power characteristics as a function of RS are shown in Fig. 11. 
At lower temperatures VOC and VPP are higher. As a result the 
I-V characteristic is shifted to the right; the output power at 
the maximum power point and FF are also higher. There is 
also a small increase in IPP (Fig. 12).  

The model of simplified PV panel is compared to the PV 
panel model with full diode description. The temperature 
dependence of the I‑V characteristic is shown in Fig. 13. 

Fig. 6. IPP, VPP and FF characteristics as a function of RS: IPP(RS); 

VPP(RS) and FF(RP). 

Fig. 7. Current and power characteristics as a function of RSH: 

I(V1,RSH) and P(V1,RSH). 

Fig. 8. IPP, VPP and FF characteristics as a function of RSH: IPP(RSH); 
VPP(RSH) and FF(RSH). 

Fig. 9. Current and power characteristics as a function of KSH: 
I(V1,KSH) and P(V1,KSH). 

Fig. 10. IPP, VPP and FF characteristics as a function of KSH: 
IPP(KSH); VPP(KSH) and FF(KSH). 

Fig. 11. Current and power characteristics as a function of the 

temperature: I(V1,T) and P(V1,T). 

Fig. 12. IPP, VPP and FF characteristics as a function of the 

temperature: IPP(T); VPP(T) and FF(T). 

The current characteristics match for all of the simulated 
temperatures. The relative difference is smaller than 2×10-6 %, 
hence it is advisable to use the simplified model in order to 
speed the simulation of large circuits, as the accuracy is not 
sacrificed. 

V. CONCLUSIONS 

A behavioral parameterized Verilog-AMS macromodel has 
been developed for PV cell and PV panel consisting of cells 
connected in series and in parallel. The model of PV panel can 
also represent a single cell. The detailed model relays on a 
Spice diode model to describe the non-linear behavior of the 
PV cell. An alternative simplified model is also provided. It 
includes only the diode equations that are important for the 
modeling of the PV behavior. It is a pure Verilog-AMS 
implementation, and does not depend on mixed language 
simulators. Both models are simulated in the Dolphin 
Integration SMASH environment. The basic PV 
characteristics are obtained from the simulation results. 

Fig. 13. Comparison between the model with Spice diode and the 
simplified model. Current and power characteristics as a function of 
the temperature: I(V1,T) and P(V1,T). 

